

ODREĐIVANJE PROPUSNE MOĆI ZATRAVLJENIH KANALA

Dr Miodrag JOVANOVIĆ
Građevinski fakultet - Beograd

REZIME

U ovom članku su prikazane dve u svetu poznate metode proračuna propusne moći zatravljenih kanala, sa težištem na računskom modelu koji obuhvata biomehaničke karakteristike vegetacije. Opisan je način kalibracije parametara ovog modela i istaknute njegove prednosti. U prilogu je dat listing originalnog kompjuterskog programa koji omogućava efikasnu primenu predložene metode u praksi.

Ključne reči: zatravljeni kanali, rapavost travnog pokrivača, koeficijenti trenja, propusna moć kanala

1. UVOD

Termin „zatravljeni kanal” odnosi se na kanal čije je korito u potpunosti obraslo travnim pokrivačem. Reč je o malim kanalima koji mogu biti prirodni vodotoci ili veštački provodnici, projektovani i izgrađeni za potrebe navodnjavanja, dreniranja površinskih i provirnih voda, ili zaštite sliva od erozije.

Travni pokrivač predstavlja ekonomičan vid oblaganja kanala, jer se erozija efikasno sprečava smanjenjem brzine toka i prigušenjem turbulencije pri dnu, kao i korenim sistemom biljke koji vezuje tlo.

S druge strane, ovakvi vodotoci se estetski dobro uklapaju u prirodni ambijent i u duhu su u svetu aktuelnog trenda "naturalne regulacije vodotoka", kao i koncepta "održivog razvoja" u odvodnjavanju gradskih područja. Nedostatak zatravljenih kanala je u potrebi njihovog sistematskog održavanja, kao i u činjenici da zauzimaju više prostora od kamenom ili betonom obloženih kanala iste propusne moći.

Za određivanje kapaciteta zatravljenog kanala bitna je procena vrednosti koeficijenta trenja. Ovim problemom se već šezdeset godina bave istraživači širom sveta. Procena uticaja vegetacije na propusnu moć korita je važna i sa stanovišta evakuacije velikih voda koje se razlivaju po inundacionim površinama.


U ovom radu će se prikazati dva u svetu poznata pristupa u rešavanju razmatranog problema, od kojih je jedan – tradicionalan, baziran na empirijskoj vezi koeficijenta trenja i proizvoda brzine i hidrauličkog radijusa, a drugi – aktuelniji, na određivanju rapavosti travnog pokrivača pomoću njegovih biomehaničkih svojstava. Oba pristupa, koja podrazumevaju jednoliko tečenje, biće ilustrovana računskim primerima.

2. KLASIČNI PRISTUP U PROCENI VREDNOSTI KOEFICIJENTA TRENJA

Tradicionalno se u hidrotehničkoj praksi koriste empirijski podaci za Manning-ov koeficijent trenja (dati u tabelarnom ili grafičkom obliku), u zavisnosti od geometrijskih karakteristika kanala, hidrauličkih parametara toka i vrste biljnog pokrivača [2]. Kod nas se najčešće koriste podaci ruskih autora [1], [3].

Na Zapadu se za zatravljene kanale koriste eksperimentalno utvrđene zavisnosti $n=f(VR)$ (Slika 1), gde je n - Manning-ov koeficijent, V - srednja profilska brzina, a R - hidraulički radijus. Ovakve zavisnosti se mogu naći u brojnoj literaturi, na primer [7].

Iako strogo uzevši, Manning-ov koeficijent zavisi od vrste trave i njene visine, oblika poprečnog preseka kanala i dubine, dugo se smatralo da na navedene zavisnosti vrlo malo utiče oblik poprečnog preseka i nagib dna kanala i da je umesto detaljnih podataka o



Sl. 1. - Vrednosti Manning-ovog koeficijenta u zavisnosti od parametra VR pri različitim „stepenu usporenja toka“ koji može biti: A - vrlo visok, B - visok, C - umeren, D - nizak, E - vrlo nizak. Može se primetiti da je korišćenje prikazanih zavisnosti, koje važe za male kanale ($VR < 2.0$), bazirano na subjektivnoj oceni „stepena usporenja“, a ne na karakteristikama biljnog pokrivača. Vrednosti koeficijenta n opadaju sa porastom vrednosti parametra VR (odnosno protoka), jer se sa povećanjem dubine smanjuje relativna rapavost
Fig. 1. „n-VR“ experimental curves for various degrees of flow retardance.

biljnoj vrsti, dovoljno uopšteno proceniti „stepen usporenja toka“ u zavisnosti od visine trave, kao što je definisano u Tabeli 1. Naknadno je ovaj stav revidovan, pa se u nekim priručnicima preciziraju geometrijske karakteristike kanala, uključujući njegov podužni nagib dna [6]. S obzirom na međusobnu zavisnost promenljivih n i V , korišćenje „n-VR“ dijagrama se obavlja u sklopu iterativnog računskog postupka, što će se ilustrirati brojnim primerom.

Tab. 1. - Tip krive određen preko „stepena usporenja toka“ [7]

Prosečna visina trave (cm)	Stepen usporenja toka	
	Gust zasad	Umereno gust zasad
>75	A	B
25 - 60	B	C
15 - 25	C	D
5 - 15	D	D
< 5	E	E

Primer 1. Odrediti protok u kanalu trapeznog preseka, širine u dnu $b=5$ m, nagiba kosina 1:2 i podužni nagiba dna $l_d=0.002$. Normalna dubina iznosi $h=0.75$ m. Visina trave je $h_v=15$ cm, a procenjenom stepenu usporenja toka odgovara kriva D na Slici 1.

(1) Na osnovu datih podataka za kanal, mogu se sračunati vrednosti površine, okvašenog obima i hidrauličkog radijusa: $A=4.88$ m², $O=8.36$ m, $R=0.58$ m.

(2) Pretpostavlja se vrednost Manning-ovog koeficijenta $n=0.04$ m^{-1/3}s, kojoj sa krive D odgovara vrednost $VR=0.3$ m²/s, pa je: $V=0.3/0.58=0.52$ m/s.

(3) S druge strane je iz jednačine jednolikog tečenja: $V'=n^{-1}R^{2/3}l_d^{1/2} = (1/0.04) \times 0.58^{2/3} \times 0.002^{1/2} = 0.78$ m/s > 0.52 m/s.

Korake (2)-(3) treba ponoviti sa novim – korigovanim vrednostima koeficijenta n , sve dok se ne slože vrednosti V i V' ; na primer, za $n=0.03$ m^{-1/3}s, dobija se $VR \approx 0.6$, $V=1.03$ m/s, $V'=0.89$ m/s; za $n=0.037$ m^{-1/3}s, $VR \approx 0.49$, $V=0.84$ m/s, $V'=0.84$ m/s, što je konačno rešenje.

(4) Vrednost protoka je: $Q = (1/0.037) \times 4.88 \times 0.58^{2/3} \times 0.002^{1/2} = 4.1$ m³/s

Osnovni nedostaci izloženog pristupa su:

- Zavisnosti $n=f(VR)$ nisu uvek jednoznačne.
- Ekstrapolacija van eksperimentalnih uslova, naročito u slučaju relativno malih nagiba dna (ispod 1%), daje nepouzdan rezultate.
- Kriterijum "stepena usporenja" je subjektivan, jer nije baziran na biomehaničkim svojstvima vegetacije.

3. RAČUNSKI MODEL ZASNOVAN NA BIOMEHANIČKIM SVOJSTVIMA TRAVE

Nije sporno da je fizički utemeljen matematički model tečenja vode preko biljnog pokrivača veoma teško realizovati. Međutim, ako se analiza svede na manji broj fizički ključnih parametara, može se doći do relativno jednostavnog modela koji daje inženjerski upotrebljive rezultate, a superioran je u odnosu na klasični pristup. Kao kod tečenja po čvrstoj podlozi, ovakav model se mora bazirati na definisanju relativne rapavosti.

Dimenzionom analizom se može pokazati da koeficijent tangencijalnog napona za tokove sa dubinom koja je značajno veća od visine biljnog pokrivača na dnu, primarno zavisi od tri bezdimenziona broja [9]:

$$C_\tau = \phi \left\{ \frac{h}{(MEI/\tau_o)^{1/4}}, \frac{k_v}{h_v}, \frac{k_v}{h} \right\} \quad (1)$$

gde je C_τ - koeficijent tangencijalnog napona [-], k_v - apsolutna rapavost trave [m], h_v - prosečna visina trave u uspravnom položaju [m], h - normalna dubina [m], MEI - parametar globalne krutosti trave [N m²], a τ_o - tangencijalni napon na dnu [Pa]. Umesto tangencijalnog napona, može se alternativno koristiti smičuća brzina $u \cdot \sqrt{\tau_o/\rho} = \sqrt{g \cdot h \cdot l_d}$, gde je ρ - gustina vode, l_d - podužni nagib dna, a g - gravitaciono ubrzanje.

Apsolutna rapavost trave k_v se teško definiše, jer zavisi od stepenja njene povijenosti pod uticajem toka. Zbog toga je uveden parametar „relativnog povijanja“ k_v/h_v ; na primer, vrednost ovog parametra 0.5 označava da je list trave savijen do polovine svoje dužine u uspravnom položaju.

Relativna rapavost k_v/h se takođe menja sa povijanjem trave i zavisi od njenih biomehaničkih svojstava, od kojih su najbitnija: visina, krutost i gustina.

Parametar MEI odražava globalnu krutost biljnog pokrivača. Naime, pokazalo se da je otpornost trave na deformaciju pod smičućim dejstvom toka jedino moguće globalno definisati jednim parametrom koji predstavlja proizvod gustine M (broj listova trave po jedinici površine) i krutosti pojedinačnog lista EI , gde je E - modul elastičnosti [Pa], a I - drugi moment površine poprečnog preska [m^4]. Najosetljiviji problem u razradi računskog modela je određivanje vrednosti parametra MEI , jer se to, kao što ćemo videti u nastavku, može obaviti jedino eksperimentalnim putem. Ustanovljena korelacija između parametra MEI i prosečne visine trave ima ovakav oblik [9]:

$$MEI = \begin{cases} \text{sveža: } 319 \cdot h_v^{3.30} \\ \text{uvela: } 25.4 \cdot h_v^{2.26} \end{cases} \quad [N \times m^2] \quad (2)$$

i pokazuje da se globalna krutost povećava sa visinom trave. Ovaj naizgled paradoksalan zaključak objašnjava se time što se povijanjem listova povećava gustina travnog pokrivača, a ovaj efekat prevladava efekat manje pojedinačne krutosti dužih travnih listova (povećanje vrednosti faktora M kompenzuje smanjenje vrednosti faktora EI).

Veza između apsolutne rapavosti, biomehaničkih karakteristika trave i hidrauličkog režima je takođe empirijskog karaktera [8], [9]:

$$k_v = 0.14 h_v \left[\frac{(MEI/\tau_0)^{0.25}}{h_v} \right]^{1.59} \quad (3)$$

i predstavlja vezu članova na desnoj strani opšteg izraza (1). Funkcija ϕ u istom izrazu može se tada predstaviti u obliku logaritamske zavisnosti između Darcy-Weisbach-ovog koeficijenta trenja (λ) i relativne rapavosti:

$$\frac{1}{\sqrt{\lambda}} = A_v + B_v \log(h/k_v), \quad (4)$$

gde su A_v i B_v empirijski parametri. Manning-ov koeficijent rapavosti se po želji može sračunati koristeći poznatu vezu:

$$n = \frac{\sqrt{\lambda}}{8g} h^{1/6}. \quad (5)$$

Parametri A_v i B_v zavise od veličine tangencijalnog napona i s tim u vezi treba ukazati na jedan interesantan prirodni fenomen. Savijanje trave pod vodom je slično „talasanju“ trave ili žita pod uticajem vetra; kada brzina toka dostigne određenu kritičnu vrednost, polegla trava formira relativno „glatku“ kompaktnu podlogu, tako da se sa daljim povećanjem brzine, tangencijalni napon pri dnu (odnosno smičuća brzina), vrlo malo menja. Ova pojava se objašnjava tako što se listovi trave svojim položajem „prilagođavaju“ strujnicama, tako da se u globalu efekat povećanja brzine kompenzuje smanjenjem apsolutne rapavosti. (Karakter „ n -VR“ krivih takođe ukazuje na ovaj fenomen). Svaka biljna vrsta dakle, ima određenu konstantnu vrednost „kritične smičuće brzine“ u_{*c} [9]:

$$u_{*c} = \min \begin{cases} 0.028 + 6.33 MEI \\ 0.23 MEI^{0.106} \end{cases} \quad [m/s], \quad (6)$$

pri čemu se prvi izraz odnosi na nisku travu koja se u potpunosti vraća u uspravan položaj (elastična deformacija), a drugi, na vegetaciju sa dugim i krutim listovima, koja se ne vraća u prvobitni uspravan položaj (zaostala, plastična deformacija). U proračunima se uzimamanjavrednost.

Vrednosti parametara A_v i B_v , eksperimentalno definisane u zavisnosti od odnosa stvarne i kritične smičuće brzine, date su u Tabeli 2.

Tab. 2. - Vrednosti parametara A_v i B_v [9]

Položaj lista trave	Kriterijum	Vrednost parametara	
		A_v	B_v
uspravan	$u^*/u_{*c} \leq 1.0$	0.15	1.85
povijen	$1.0 < u^*/u_{*c} \leq 1.5$	0.2	2.70
povijen	$1.5 < u^*/u_{*c} \leq 2.5$	0.28	3.08
povijen	$2.5 < u^*/u_{*c}$	0.29	3.50

Sa sračunatim vrednostima koeficijenta trenja, propusna moć zatravljenog kanala može se odrediti iterativnim postupkom poput onog kod „ n -VR“ metode. U prilogu je dat listing originalnog kompjuterskog programa koji omogućava brzo rešavanje datog problema.

Kalibracija parametara modela

Mehanička svojstva trave se teško mogu direktno utvrditi, već se posredno definišu poređenjem sa poznatim mehaničkim svojstvima veštačke trave (plastične podloge sa savitljivim elementima različite visine i

gustine), koja se u strujnom polju ponaša slično prirodnoj travi¹. Nekoliko činjenica potvrđuje da je moguće uspostaviti korelaciju u hidrauličkom smislu između prirodne i veštačke trave:

- Vizuelno se može uočiti sličnost strujnog polja, pri čemu je, kao što je već rečeno, presudan globalni uticaj, a ne uticaj svojstava pojedinačnih elemenata rapavosti.
- U oba slučaja se raspored brzine po dubini toka može opisati logaritamskom krivom [10].
- Sa veštačkim materijalom se mogu konstruisati „n-VR” krive slične onim koje su eksperimentalno utvrđene za prirodnu vegetaciju [8].

Vrednosti parametra *MEI* za veštačke podloge su kalibrisane primenom opisanog računskog modela, koristeći „n-VR” krive za razne biljne vrste [8], [9]. Izrazi (2), (3) i (6) su odraz uopštavanja rezultata te kalibracije. Brojne vrednosti parametara *A_v* i *B_v* su takođe rezultat eksperimentalnog rada sa veštačkim podlogama, jer se jedino na taj način mogu pouzdano utvrditi pojedini biomehanički parametri i izvoditi ponovljivi eksperimenti.

Spomenimo i to da se određivanje vrednosti parametra *MEI* može lako obaviti i terenskim postupkom koji ne zahteva eksperimente u laboratorijskim kanalima. Ova postupak, poznat kao „test sa pločom”, sastoji se u tome da se ploča određenih dimenzija (1829 × 305 mm) i određene mase (4.85 kg) pusti da iz vertikalnog položaja slobodno padne na sloj trave, rotirajući oko kraće strane koja je u kontaktu sa tlom. Ploča pri padu klizi po travi i deluje na nju smičućom silom koja povija travu na sličan način kao vodeni tok. Najveće vertikalno odstojanje pale ploče do tla predstavlja meru otpornosti travnog pokrivača, u kojoj su sadržani kombinovani uticaji visine trave, njene gustine i krutosti. Detalji ovog postupka mogu se naći u literaturi [9].

Test sa pločom predstavlja pogodan način da se eksperimentalna baza podataka na kojoj počiva kalibracija parametara modela stalno proširuje kako na biljne vrste koje do sada nisu ispitivane, tako i na one za koje postoje oskudni, ili nedovoljno pouzdani podaci.

Primer 2. Odrediti propusnu moć kanala iz prethodnog primera, ako se pretpostavi da je prosečna visina lista trave 15 cm. (Ulazni podaci su dati u prilogu uz listing programa).

¹Eksperimentalno korišćenje veštačke umesto prirodne trave ima isti smisao kao korišćenje „ekvivalentne peščane rapavosti” u vezi sa poznatim Nikuradse-ovim eksperimentima tečenja pod pritiskom u cevima. U tom kontekstu se parametar *MEI* može shvatiti kao „ekvivalentna krutost plastike”.

a) Osnovni rezultati:

Propusna moc zatravljenog kanala

Podaci o kanalu:

Oblik poprečnog preseka:	trapezni
Širina u dnu (m):	5.00
Nagib kosine 1:	2.00
Poduzni nagib korita:	0.0020

Podaci o travi:

Vrsta trave:	sveza
Visina trave (m):	0.15

Q (m ³ /s)	h (m)	A _v (m ²)	B (m)	O (m)	R (m)	V (m/s)	Fr (-)	n (-)
2.00	0.52	3.12	7.07	7.31	0.43	0.64	0.095	0.039
5.00	0.75	4.88	8.00	8.36	0.58	1.02	0.175	0.030
8.00	0.92	6.29	8.68	9.11	0.69	1.27	0.227	0.027
10.00	1.01	7.13	9.06	9.54	0.75	1.40	0.255	0.026
15.00	1.21	9.02	9.86	10.43	0.87	1.66	0.308	0.024

b) Pomoćni rezultati:

Krutost <i>MEI</i> (N m ²):	0.6094
Kritična smičuća brzina <i>u*_c</i> (m/s):	0.2182


Q (m ³ /s)	h (m)	u* (m/s)	A _v (-)	B _v (-)	kv (m)	λ (-)	n (-)	nit (-)
2.00	0.52	0.101	1.85	0.15	0.140	0.152	0.039	16
5.00	0.75	0.121	1.85	0.15	0.121	0.080	0.030	16
8.00	0.92	0.134	1.85	0.15	0.112	0.061	0.027	16
10.00	1.01	0.141	1.85	0.15	0.107	0.054	0.026	16
15.00	1.21	0.154	1.85	0.15	0.100	0.044	0.024	16

Analiza promene rapavosti i vrednosti koeficijenata trenja prikazana je na Slici 2.


Imajući u vidu način na koji je izvršena kalibracija parametara, postavlja se opravdano pitanje u čemu je prednost opisanog računskog modela u odnosu na klasičnu „n-VR” metodu? Izvesno je da u domenu podataka na osnovu kojih je izvršena kalibracija, fizički baziran model ne daje veću tačnost od klasične „n-VR” metode. Međutim, prednosti ovog modela su nesporne:

- Omogućava uvid u fizičke aspekte pojave i u relativni uticaj biomehaničkih svojstava biljnog pokrivača na hidrauličke karakteristike toka, a analizom osetljivosti parametara, može se proceniti uticaj neizvesnosti pojedinih ulaznih podataka na rezultate proračuna.

(a)



(b)



Slika 2: Uticaj tangencijalnog napona na: (a) apsolutnu i relativnu rapavost travnog pokrivača; (b) vrednosti koeficijenta trenja. Može se primetiti da vrednosti ovih koeficijenata rastu sa povećanjem prosečne visine trave i globalnog parametra njene krutosti. Computational results showing dependence of the absolute/relative roughness and friction coefficients on the bottom shear stress for a particular canal and type of grass.

- U odnosu na „n-VR” metodu, ima opštiji karakter jer omogućava tačniju ekstrapolaciju na oblast nagiba dna ispod 1%, kao i dobijanje pouzdanijih rezultata tamo gde se javljaju teškoće u primeni „n-VR” dijagrama, bilo usled subjektivnosti u izboru merodavne krive, ili usled rasipanja eksperimentalnih podataka.
- Budući eksperimenti u prirodnim i laboratorijskim uslovima (pre svega relativno jednostavan test sa pločom), mogu samo doprineti opštosti modela, ne remeteći njegovu postojeću osnovnu strukturu.
- Opisani računski postupak se lako može integrisati u složenije modele, kakav je na primer onaj za procenu propusne moći korita sa inudacijama obraslim različitim vrstama vegetacije [4], [5].
- Model omogućava analizu stabilnosti korita obraslih travom, nezavisno od svih dosadašnjih kriterijuma, a parametar MEI se može koristiti za izbor travne obloge kanala (ako postoji mogućnost izbora, tj. ako nisu presudni klimatološki ili neki drugi faktori).

4. MAKSIMALNA BRZINA U ZATRAVLJENIM KANALIMA

Maksimalna brzina toka u zatravljenim kanalima sa stanovišta opstanka biljnog pokrivača, zavisi od vrste trave, vrste tla i podužnog nagiba kanala. U proceni ove brzine, opisani računski model dolazi do punog izražaja, u kombinaciji sa postojećim kriterijumima. Ilustracije radi, vrednosti maksimalne brzine za neke tipove travnih obloga date su u Tabeli 3. U projektovanju zatravljenih kanala takođe važi iskustvena preporuka da maksimalna brzina toka ne bude 120-130% veća od maksimalne (dozvoljene) brzine u odgovarajućim neobloženim kanalima (bez biološke zaštite).

Ta b . 3 . - Maksimalne brzine u zatravljenim kanalima [7]

Vrsta trave	Podužni nagib kanala (%)	Maksimalna brzina (m/s)	
		slabo erodibilno tlo	erodibilno tlo
Zubača (troskot, 0 - 5	2.4	1.8	
Bermudska trava)	5 - 10	2.1	1.5
(<i>Cynodon dactylon</i>)	>10	1.8	1.2
Livadarka	0 - 5	2.1	1.5
(Bafalo trava)	5 - 10	1.8	1.2
(<i>Buchlo dactyloides</i>)	>10	1.5	0.9
Mešavine	0 - 5	1.5	1.2
	5 - 10	1.2	0.9
	>10	ne preporučuje se	
Detelina	0 - 5	1.1	0.8
	(<i>Alfalfa</i>)	>5	ne preporučuje se

5. ZAKLJUČCI

1. Propusna moć zatravljenih kanala može se odrediti tradicionalnim postupkom pomoću takozvanih „n-VR” krivih, ili pomoću fizički zasnovanog računskog modela.
2. Računski model, baziran na konceptu relativne rapavosti, ima niz prednosti u odnosu na klasični pristup, od kojih su najbitnije, da pruža uvid u fizičke aspekte problema, da ima univerzalniji domen primene i da nudi mogućnost dalje nadgradnje.
3. Pomoću opisanog modela moguće je izvršiti izbor optimalne biološke zaštite kanala, kao i procenu stabilnosti korita u uslovima velikih voda.

LITERATURA

- [1] Altunin, S.T.: *Regulirovanie rusel*, Seljhozgiz, Moskva, 1956.
- [2] Chow, V.T.: *Open-Channel Hydraulics*, McGraw-Hill, New York, 1959.
- [3] Čertousov, M., D.: *Gidravlika*, Gosenergoizdat, Moskva, 1957.
- [4] Darby, S., D., Thorne, C.: *Predicting Stage-Discharge Curves in Channels with Bank Vegetation*, Journal of Hydraulic Engineering, ASCE, Vol. 122, No. 10, 1996.

- [5] Darby, S., Rinaldi, M., Pistoiesi, I.: *Effects of Flexible Riparian Vegetation Growth on Discharge Capacity*, Proceedings, XXVII. IAHR Congress, San Francisco, 1997.
- [6] Goldman, S., J., Jackson, K., Bursztynsky, T., A.: *Erosion / Sediment Control Handbook*, McGraw-Hill, New York, 1986.
- [7] Kinori, B., Z.: *Manual of Surface Drainage Engineering*, Volume I, Elsevier, Amsterdam, 1970.
- [8] Kouwen, N., Li, R., M.: *Biomechanics of Vegetative Channel Linings*, Journal of the Hydraulics Division, ASCE, Vol. 106, No. HY6, 1980.
- [9] Kouwen, N.: *Field estimation of the biomechanical properties of grass*, Journal of Hydraulic Research, Vol. 26, No. 5, 1988.
- [10] Temple, D., M.: *Closure of Velocity Distribution Coefficients for Grass Lined Channels*, Journal of the Hydraulics Division, ASCE, Vol. 113, No. 9, 1987.

DISCHARGE CAPACITY EVALUATION IN VEGETATED CHANNELS

by

Miodrag JOVANOVIĆ, Ph.D
Faculty of Civil Engineering, University of Belgrade

Summary

Two existing methods for discharge capacity evaluation in vegetated canals are presented in this article, the emphasis being on the model based on the biomechanical properties of vegetative canal lining. The calibration of model parameters is discussed, as well as the advantages of this model. The listing of an original

computer code is provided in order to facilitate the application of the suggested method in practice.

Key words: vegetated canals, resistance of grass linings, friction coefficients, canal discharge capacity

Redigovano 30.06.1999.

P R I L O G

```

-----
! Program:   TravaQh
! Funkcija:  Propusna moc zatravljenog kanala
! Opis:     Korito prizmaticno,presek trapezni
! Metoda:   Iterativna, polovljenje intervala
! Autor:    M. Jovanovic, 19.12.1998.
! Okruzenje: Lahey Fortran90, LF90,V.2.00a,'95.
!           Microsoft Power Station,V.4, 1995.
-----

```

```

MODULE karakteristike_kanala
  implicit none
  TYPE :: kanal_tip
    character(10) :: oblik_pop_preseka
    real :: sirina ! sirina u dnu
    real :: kosina ! kotangens ugla kosine
    real :: pad ! poduzni nagib dna
    real :: A ! površina preseka
    real :: Bn ! sirina nivoa
    real :: O ! okvaseni obim
    real :: R ! hidraulički radijus
    real :: AR23 ! funkcija  $A \cdot R^{2/3}$ 
  END TYPE kanal_tip
  TYPE(kanal_tip) :: kanal
  contains

  SUBROUTINE funkcije_pop_preseka(kanal,h)
    implicit none
    TYPE(kanal_tip), intent(inout) :: kanal
    real, intent(in) :: h
    kanal%A=h*(kanal%sirina+kanal%kosina*h)
    kanal%Bn=kanal%sirina+2.*h*kanal%kosina
    kanal%O =kanal%sirina+
      & 2.*h*SQRT(kanal%kosina**2+1.)
    kanal%R = kanal%A / kanal%O
    kanal%AR23 = kanal%A * kanal%R **.6667
  END SUBROUTINE funkcije_pop_preseka
END MODULE karakteristike_kanala
-----

```

```

MODULE karakteristike_trave
  implicit none
  TYPE :: trava_tip
    character(5) :: vrsta
    real :: visina ! visina trave
    real :: MEI ! sumarna krutost
    real :: uzv_c ! krit. smicuca brzina
    real :: Av, Bv ! konstante
  END TYPE trava_tip
  TYPE(trava_tip) :: trava
  contains

  SUBROUTINE krutost(trava)
    implicit none
    TYPE(trava_tip), intent(inout) :: trava
    if(trava%vrsta == "sveza") then
      trava%MEI = 319. * trava%visina**3.3
    elseif(trava%vrsta == "uveala") then
      trava%MEI = 25.4 *trava%visina**2.26
    else
      stop "Greska u proracunu krutosti !"
    endif
    trava%uzv_c=MIN(0.028+6.33*trava%MEI, &
      & 0.23*trava%MEI**0.106 )
  END SUBROUTINE krutost

  SUBROUTINE konstante(trava,uzv)
    implicit none
    TYPE(trava_tip), intent(inout) :: trava
    real, intent(in) :: uzv
    real :: uu
    uu = uzv / trava%uzv_c
    if(uu <= 1.) then
      trava%Av = 1.85; trava%Bv = 0.15
    elseif(uu > 1. .and. uu <= 1.5) then
      trava%Av = 2.70; trava%Bv = 0.20
    elseif(uu > 1.5 .and. uu <= 2.5) then
      trava%Av = 3.08; trava%Bv = 0.28
    elseif( uu > 2.5 ) then

```

```

        trava%Av = 3.50;  trava%Bv = 0.29
    endif
END SUBROUTINE konstante

END MODULE karakteristike_trave
-----

MODULE trenje
    implicit none
    real :: tau, kv

    TYPE :: koef_trenja_tip
        real :: lambda
        real :: Manning
    END TYPE koef_trenja_tip
    TYPE(koef_trenja_tip) :: koef_trenja
contains

SUBROUTINE koef_trenja_racun      &
    (koef_trenja, h, uzv)
    USE karakteristike_trave
    implicit none
    TYPE(koef_trenja_tip),      &
    intent(inout):: koef_trenja
    real, intent(in)  :: h, uzv
    real, parameter :: g = 9.81, ro = 1000.
    tau = ro * uzv * uzv
    kv = 0.14 * trava%visina *      &
    ( (trava%MEI / tau) ** 0.25 /      &
    trava%visina ) ** 1.59
    koef_trenja%lambda = 1./(trava%Av * &
    LOG( h / kv ) + trava%Bv ) ** 2
    koef_trenja%Manning = SQRT
    ( koef_trenja%lambda /8./g ) * h**.1667
END SUBROUTINE koef_trenja_racun

END MODULE trenje
-----

!-----
MODULE racun
    real, allocatable, dimension(:) :: Qu
    real :: Q
contains

FUNCTION fh(h)
    USE karakteristike_kanala
    USE karakteristike_trave; USE trenje
    real, parameter :: g = 9.81
    real :: fh, uzv
    uzv = SQRT( g * h * kanal%pad )
    CALL konstante(trava, uzv)
    CALL koef_trenja_racun(koef_trenja,h,uzv)
    CALL funkcije_pop_preseka(kanal, h)
    fh = (koef_trenja%Manning * Q /
           kanal%AR23)**2 - kanal%pad
END FUNCTION fh

SUBROUTINE neli(xd,xg,f,xo,eps,nit,ier)
    implicit none
    real, external  :: f
    real, intent(in)  :: xd, xg, eps
    real, intent(out) :: xo
    integer, intent(out) :: nit, ier
    integer          :: it
    real :: xa, xb, xp, yd, yg, yo, p, po
    xa = xd;  xb = xg;  xp = xa
    if( xa >= xb ) then
        xa = xb;  xb = xp
    endif
    yd = f(xa);  yg = f(xb)
    p = yd * yg
    if( p == 0. ) then
        if( yd == 0. ) xo = xa
        if( yg == 0. ) xo = xb
        nit = 1; ier = 0; return
    elseif( p < 0. ) then
        nit=INT( LOG( (xb-xa)/eps) /LOG(2.)) +1

```



```

do it = 1, nit
  xo = 0.5 * (xa + xb); yo = f(xo)
  po = yd * yo
  if( po == 0. ) then
    nit = it; ier = 0; return
  elseif( po < 0. ) then
    xb = xo; yg = yo
  else
    xa = xo; yd = yo
  endif
enddo

xo = 0.5 * (xa + xb)
nit = it; ier = 0; return
else
  ier = 1; return
endif
END SUBROUTINE neli

END MODULE racun
!-----
SUBROUTINE ucitavanje_podataka(nq)
  USE karakteristike_kanala
  USE karakteristike_trave; USE racun
  integer, parameter :: nqmax = 50
  integer             :: nq
  character(12)       :: datio, datu,      &
                        dati1, dati2?
  character(80)       :: text
  NAMELIST/podaci_kanal/ kanal
  NAMELIST/podaci_trava/ trava

  print "(5x,A)", &
  "Ime fajla sa imenima I/O fajlova? "
  read "(A)", datio
  open(UNIT=10, FILE=datio, STATUS="old")

  read(10,"(A)") datu ! ulazni podaci
  read(10,"(A)") dati1 ! osnovni rezultati
  read(10,"(A)") dati2 ! pomocni rezultati
  open(UNIT=1, FILE=datu , STATUS="old")
  open(UNIT=2, FILE=dati1, STATUS="replace")
  open(UNIT=3, FILE=dati2, STATUS="replace")

  read(1,fmt="(A)") text
  read(1, NML = podaci_kanal)
  read(1, NML = podaci_trava)

  read(1,*) nq
  if(nq > nqmax) stop 'Previše podataka Qu!'
  ALLOCATE( Qu(nq) )
  read(1,*) Qu

  close(UNIT=1)

  write(2,fmt="(A)") text
  write(3,fmt="(A)") text
  write(2,fmt="( /2x,A, /2x,2A,2 (/2x,A,f6.2) , &
  /2x,A,f6.4) " ) &
  "Podaci o kanalu:", &
  "Oblik poprecnog preseka: ", &
  kanal%oblik_pop_preseka, &
  "Sirina u dnu (m): ", kanal%sirina, &
  "Nagib kosine 1: ", kanal%kosina, &
  "Poduzni nagib korita: ", kanal%pad
  write(2,fmt="( /2x,A/2x,2A/2x,A,f6.2) " ) &
  "Podaci o travi:", &
  "Vrsta trave: ", trava%vrsta, &
  "Visina trave (m): ", trava%visina
END SUBROUTINE ucitavanje_podataka

```

```

PROGRAM TravaQh
  USE karakteristike_kanala
  USE karakteristike_trave
  USE trenje; USE racun
  integer      :: nq, nit, ier = 0
  real, parameter :: g = 9.81, eps = 0.001
  real :: hmin = .001, hmax = 20., h, V, Fr

  CALL ucitavanje_podataka(nq)
  CALL krutost(trava)
  write(3,103) trava%vrsta, trava%visina, &
    trava%MEI, trava%uzv_c
  write(3,101); write(2,100)

  do i = 1, nq
    Q = Qu( i )
    CALL neli(hmin,hmax,fh,h,eps,nit,ier)
    if(ier /= 0) stop
    'Nema resenja za normalnu dubinu!'
    uzv = SQRT( g * h * kanal%pad )
    CALL funkcije_pop_preseka(kanal, h)
    V = Q / kanal%A
    Fr = Q * Q * kanal%Bn / g /kanal%A ** 3
    write(2, fmt =
      "(2f7.2,3f8.2,2f7.2,2f8.3)" )
    Q,h,kanal%A,kanal%Bn,kanal%O,kanal%R, &
    V, Fr, koef_trenja%Manning
    write(3, fmt=
      "(2f7.2,f8.3,2f7.2,3f8.3,i5)" )
    Q,h,uzv,trava%Av,trava%Bv,kv, &
    koef_trenja%lambda,koef_trenja%Manning,&
    nit
  enddo

```

```

100 format(//4x,'Q',6x,'h',7x,'A',7x,'B',7x, &
'O',6x,'R',6x,'V',6x,'Fr',6x,'n'/1x,'(m3/s)', &
3x,'(m)',5x,'(m2)',4x,'(m)',5x,'(m)',4x,'(m)', &
3x,'(m/s)',4x,'(-)',4x,'(..)'/)

101 format(//4x,'Q',6x,'h',6x,'u*',6x,'Av',5x,&
'Bv',5x,'kv',4x,'lambda',4x,'n',5x,'nit'/1x, &
'(m3/s)',3x,'(m)',4x,'(m/s)',4x,'(-)',4x,'(-)' &
,4x,'(m)',4x,'(-)',5x,'(..)',3x,'(-)'/)

103 format(//5x,'Vrsta trave: ',A, &
/5x,'Visina (m): ',f6.2, &
/5x,'Krutost MEI (N m2): ', f8.4, &
/5x,'Kriticna smicuca brzina u*_c (m/s): ', &
f8.4/ )

close(UNIT=2); close(UNIT=3)

END PROGRAM TravaQh

```

Ulazni podaci:

Propusna moc zatravljenog kanala

&podaci_kanal kanal = "trapezni",
5., 2., .002 /

&podaci_trava trava = "sveza", 0.15 /
5 nq - broj protoka
2 5 8 10 15 Qu - vrednosti protoka

Napomena: Autor ne snosi odgovornost za posledice korišćenja ovog programa koji je izraden za potrebe rešavanja jednog konkretnog problema iz prakse.